
[bookmark: _Toc7508548]6.2.9	Communication port types
Ports facilitate communication between test components and between test components and the test system interface.
TTCN‑3 supports message-based and procedure-based ports. Each port shall be defined as being message-based or procedure-based. Message-based ports shall be identified by the keyword message and procedure-based ports shall be identified by the keyword procedure within the associated port type definition.
[bookmark: _GoBack]Ports are bidirectional. The directions are specified by the keywords in (for the in direction), out (for the out direction) and inout (for both directions). Operations allowed on a procedure present in the incoming port list are getcall, reply and raise. Operations allowed on a procedure present in the outcoming port list are call, getreply and catch. Directions shall be seen from the point of view of the test component owning the port with the exception of the test system interface, where directions shall be seen from the point of view of the test component port mapped to the test system interface port. The in list of the test system interface port contains message or procedure for which the mapped test component port allows the following operations: receive, trigger, getcall, reply or raise. The out list of the test system interface port contains message or procedure for which the mapped test component port allow the folowing operations: send, call, getreplay or catch.. Operations allowed on a procedure present in the incoming port list are getcall, reply and raise. Operations allowed on a procedure present in the outcoming port list are call, getreply and catch.
Each port type definition shall have one or more lists indicating the allowed collection of (message) types or procedure signatures together with the allowed communication direction.
For configuration purposes the port type may have one map param and one unmap param declaration indicating the allowed additional parameters for the respective operation. These formal parameters shall be value parameters.
Whenever a signature (see also clause 14) is defined in the out direction of a procedure-based port, the types of all its inout and out parameters, its return type and its exception types are automatically part of the in direction of this port. Whenever a signature is defined in the in direction for a procedure-based port, the types of all its inout and out parameters, its return type and its exception types are automatically part of the out direction of this port.
Ports used for the communication with the SUT may need to address specific entities within the SUT. In addition, several address schemes may be supported by one SUT at different ports. To support such addressing schemes, TTCN-3 allows to bind an address type to a port. Values of this type may be used for addressing purposes in communication operations (see clause 22.1) and be stored in variables. The handling of address types bound to different ports by means of the dot notation is explained in clause 6.2.12.
Syntactical Structure
Message-based port:
type port PortTypeIdentifier message "{"
		{ (address Type ";") |
		  (map param "(" { FormalValuePar [","] }+ ")") |
		  (unmap param "(" { FormalValuePar [","] }+ ")") |
		  ((in | out | inout) { MessageType [ "," ] }+ ";") } 
"}"

Procedure-based port:
type port PortTypeIdentifier procedure "{"
		{ (address Type ";" ) |
		  (map param "(" { FormalValuePar [","] }+ ")") |
		  (unmap param "(" { FormalValuePar [","] }+ ")") |
		  ((in | out | inout) { Signature [ "," ] }+ ";") } 
"}"

TTCN-3 allows to define constants, variables and parameters of a port type. These constants, variables or parameters can contain a reference to an existing component port or a special value null. The special value null represents an unspecified port reference, i.e. it can be used to explicitly allow the referencing of no port.
Port type values are object references and follow specific rules for this kind of values.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
a)	At most one address type shall be bound to a port type.
b)	At most one map parameter list shall be defined for a port type.
c)	At most one unmap parameter list shall be defined for a port type.
d)	Formal parameters of map param and unmap param declarations shall be value parameters and not be of port, component, timer or default type or of structured types having fields of port, component, timer or default type.
e)	MessageType shall be a data type as defined in clause 3.1.
Examples
EXAMPLE 1:	Message-based port
	// Message-based port which allows types MsgType1 and MsgType2 to be received at, MsgType3 to be
	// sent via and any integer value to be send and received over the port 
	type port MyMessagePortTypeOne message 
	{
		in		MsgType1, MsgType2;
		out		MsgType3;
		inout 	integer
	}

EXAMPLE 2:	Procedure-based port
	// Procedure-based port which allows the remote call of the procedures Proc1, Proc2 and Proc3.
	// Note that Proc1, Proc2 and Proc3 are defined as signatures
	type port MyProcedurePortType procedure 
	{
		out		Proc1, Proc2, Proc3
	}

EXAMPLE 3:	Message-based port with address type definition
	type port MyMessagePortTypeTwo message 
	{
		address	integer;		// if addressing is used on ports of type MyMessagePortTypeTwo 
 							    // the addresses have to be of type integer
		inout	MsgType1, MsgType2;
	}

NOTE:	The term message is used to mean both messages as defined by templates and actual values resulting from expressions. Thus, the list restricting what may be used on a message-based port is simply a list of type names.
EXAMPLE 4:	Usage of param in port declaration
	// Message based port which allows MsgType4 to be send and received over the port
	// and MsgType5 and MsgType6 as configuration parameter type
	type port MyMessagePortType message 
	{
		inout	MsgType4;
		map param	(in MsgType5 p_p1, out MsgType6 p_p2);
	}

	// Procedure based port which allows the remote call of the procedure Proc1
	// and MsgType5 as configuration parameter type
	type port MyProcedurePortType procedure 
	{
		out		Proc1;
		unmap param	(MsgType5 p_p1);
	}

